Contents

I Trails Covering the Basics of SaC 3

1 Running the first program 4
  1.1 A Checklist ........................................... 4
  1.2 Create your first SaC Source File ....................... 4
  1.3 Compile the Source File and Run the Program ........ 5

2 Array Programming Basics 6
  2.1 Lesson: Arrays as Data ................................ 6
    2.1.1 Defining Arrays .................................. 6
    2.1.2 Arrays and Variables .............................. 9
  2.2 Lesson: Shape-Invariant Programming .................. 11
    2.2.1 Standard Array Operations ....................... 11
    2.2.2 Axis Control Notation ............................ 19
    2.2.3 Putting it all Together ........................... 25

3 Basic Program Structure 28
  3.1 Lesson: Functions and their Types ..................... 28
    3.1.1 Function Definitions ............................. 28
    3.1.2 Built-in Types and Subtyping .................... 29
  3.2 Lesson: Function Bodies ............................... 30
    3.2.1 Variable Declarations ............................ 30
    3.2.2 Assignments ...................................... 30
    3.2.3 Conditionals ..................................... 31
    3.2.4 Loops ............................................ 32
    3.2.5 Explicit Control Flow Manipulation ............... 33
  3.3 Lesson: Advanced Topics ............................... 33
    3.3.1 User-defined Types ............................... 33
    3.3.2 Type Conversions ................................ 34
    3.3.3 Function Overloading ............................. 34

4 With-Loops 36
  4.1 Lesson: With-Loop Basics .............................. 36
    4.1.1 Basic Components ................................. 36
    4.1.2 Generator Ranges ................................ 38
CONTENTS

4.1.3 Generator Expressions .................................... 39
4.1.4 Reductions and further With-Loop Operations ...... 40

5 Case Studies .......................................................... 43
5.1 Lesson: Image Processing ....................................... 43
5.2 Lesson: Computing Mandelbrot Images .................... 47
Part I

Trails Covering the Basics of SaC
Chapter 1

Running the first program

The following instructions will help you write your first SaC program.

1.1 A Checklist

To successfully write and run your first SaC program, you will need:

- An **ANSI C compiler**, such as gcc. Though not needed directly, the SaC compiler relies on it.

- The **SaC compiler** sac2c. It can be downloaded at <http://www.sac-home.org/>. For convenience, you should make sure, that sac2c is located in your $PATH.

- The **SaC standard library** that comes with sac2c as part of the SaC distribution. To make sure sac2c will be able to find it, the environment variable $SACBASE should point to it, i.e., ls $SACBASE should yield at least the subdirectory stdlib.

- Your favorite **text editor**, such as vi or emacs.

1.2 Create your first SaC Source File

Start your editor and type the following program:

Listing 1.1: Hello World

```c
use StdIO: all;
use Array: all;

int main()
{
    printf( "Hello World!\n");
    return(0);
}
```
As you can see, it has a strong resemblance to C. The major difference are
the module use declarations at the beginning of the program. For now, it suffices
to keep in mind, that these two use declarations for most experiments will do
the job.

In order to proceed, save this program into a file named world.sac.

1.3 Compile the Source File and Run the Program

The SaC compiler invocation is similar to the standard invocation of C compil-
ers. A typical shell session for compiling world.sac could be:

```bash
> cd /home/sbs/sac/
> ls
world.sac
> sac2c world.sac
> ls
a.out  a.out.c  world.sac
> a.out
Hello World!
>
```

The compilation process consists of two steps. First, the SaC compiler gen-
erates a C file, which then is compiled into target code by utilizing the system’s
C compiler. If no target file name is specified, the intermediate C file is named
a.out.c so that the subsequent invocation of the C compiler creates an exe-
cutable named a.out.

In the same way the default target name a.out is borrowed from standard
C compilers, the -o option for specifying target names is adopted as well. For
example, sac2c -o world world.sac results in files world.c and world.

Note here, that the compiled program, depending on the operating system, is
linked either statically or dynamically. However, it does not require any further
linking or interpretation.
Chapter 2

Array Programming Basics

This trail gives an introduction to the basic concepts of array programming in SaC. It consists of two lessons: Arrays as Data and Shape-Invariant Programming. In the former lesson, the major differences between arrays in SaC and arrays in more mainstream languages are explained. The lesson Shape-Invariant Programming gives an introduction into the most important array operations available in SaC. Based on these operations, several small examples demonstrate how more complex array operations can be constructed by simply combining the basic ones.

2.1 Lesson: Arrays as Data

In SaC, arrays are the only data structures available. Even scalar values are considered arrays. Each array is represented by two vectors, a so-called shape vector and a data vector. An array’s shape vector defines its shape, i.e., its extent within each axis, and its dimensionality (or rank), which is given implicitly by the shape vector’s length.

The section on Defining Arrays explains how arrays of various dimensionality can be defined in SaC, and how they can be generated via nesting. Furthermore, some elementary notation such as scalars, vectors, and matrices is defined.

The section on Arrays and Variables discusses the purely functional array model used in SaC.

2.1.1 Defining Arrays

In this section, several means for specifying arrays are explained.

In principle, all arrays in SaC can be defined by using the reshape operation. reshape expects two operands, a shape vector and a data vector, both of which are specified as comma separated lists of numbers enclosed in square brackets.

To get started, try the following program:
CHAPTER 2. ARRAY PROGRAMMING BASICS

Listing 2.1: Defining Arrays I

```c
use StdIO: all;
use Array: all;

int main()
{
    print( reshape( [5], [1,2,3,4,5]));
    print( reshape([3,2], [1,2,3,4,5,6]));
    print( reshape([3,2,1], [1,2,3,4,5,6]));
    return(0);
}
```

It prints three arrays:

- an array of dimensionality 1 with 5 elements \([1,2,3,4,5]\);
- an array of dimensionality 2 with 3 rows and 2 columns, and
- a 3-dimensional array with 3 elements in the leftmost axis, 2 elements in the middle axis, and one element in the rightmost axis.

Note here, that the function `print` can be applied to arbitrary arrays. Besides printing its argument’s dimensionality and shape, i.e., its shape vector, a more intuitive representation of the array’s data vector is shown. However, as the terminal allows for 2 dimensions only, arrays of higher dimensionality are interpreted as nestings of 2-dimensional arrays. Therefore, the 3-dimensional array is printed as a 2-dimensional array of vectors.

**Exercise 1** In all these examples, the product of the shape vector matches the length of the data vector. What do you expect to happen, if this condition does not hold?

For reasons of convenience, we use the following terminology:
- **scalar** always denotes an array of dimensionality 0,
- **vector** always denotes an array of dimensionality 1, and
- **matrix** always denotes an array of dimensionality 2.

As all arrays can be defined in terms of `reshape`, the following program as well is perfectly legal:

Listing 2.2: Defining Arrays II

```c
use StdIO: all;
use Array: all;

int main()
{
    print( reshape( [1], [1]));
    print( reshape( [], [1]));
    return(0);
}
```
The most interesting aspect of this program is the array defined in line 7. The empty shape vector makes it a 0-dimensional array, i.e., a scalar. The data vector carries the scalar’s value, which, in this example, is 1.

**Exercise 2** The arguments of `reshape` are vectors, i.e., arrays of dimensionality 1. Can they be specified by `reshape` expressions themselves?

The `reshape` notation is relatively clumsy, in particular, when being used for scalars. Therefore, scalars and vectors can alternatively be specified by shortcut notations as well.

For experimenting with these, try the following:

```
Listing 2.3: Shortcut Notation for Arrays
```
```c
use StdIO: all;
use Array: all;

int main()
{
    print( 1);
    print( [1,2,3,4,5]);
    print( [ [1,2], [3,4], [5,6]]);
    print( genarray( [4,3,2], 1));
    print( genarray( [4,3], [1,2]));
    return(0);
}
```

From these examples, we can see that scalars can be used in the same way as in most programming languages, and that the notation used for the parameters of `reshape` in the examples above in fact is a standard abbreviation of SaC. The example in line 8 shows that nestings of arrays are implicitly eliminated, i.e., the resulting array is identical to `reshape( [3,2], [1,2,3,4,5,6])`. For this reason, array nestings in SaC always have to be **homogeneous**, i.e., the inner array components have to have identical shapes.

Furthermore, a new function is introduced: `genarray`. It expects two arguments, a shape vector that defines the shape of the result and a default element to be inserted at each position of the result. As shown in the example of line 10, the individual array elements can be non-scalar arrays as well, which implicitly extends the dimensionality of the result array.

**Exercise 3** Given the language constructs introduced so far, can you define an array that would print as

```
Dimension: 3
Shape : < 5, 2, 2>
< 0 0 > < 0 0 >
< 1 0 > < 0 0 >
< 0 1 > < 0 0 >
< 0 0 > < 1 0 >
< 0 0 > < 0 1 >
```

, but whose definition does not contain the letter 1 more than once?
2.1.2 Arrays and Variables

This section explains why in SaC arrays are data and not containers for values as found in most other languages.

So far, all examples were expression based, i.e., we did not use any variables. Traditionally, there are two different ways of introducing variables. In conventional (imperative) languages such as C, variables denote memory locations which hold values that may change during computation. In functional languages, similar to mathematics, variables are considered place holders for values. As a consequence, a variable’s value can never change. Although this difference may seem rather subtle at first glance, it has quite some effects when operations on large data structures (in our case: large arrays) are involved.

Let’s have a look at an example:

Listing 2.4: Variables as Placeholders

```sac
use StdIO: all;
use Array: all;

int main()
{
    a = [1,2,3,4];
    print( a);
    b = modarray( a, [0], 9);
    print( b);
    return(0);
}
```

The function `modarray` expects three arguments: an array to be "modified", an index that indicates the exact position within the array to be "modified", and the value that is to be inserted at the specified position. As we would expect, the resulting array `b` is almost identical to `a`, only the very first element has changed into 9.

**Note here, that indexing in SaC always starts with index 0!**

Referring to the container / place holder discussion, the crucial question is: does the variable `a` denote a container, whose value is changed by `modarray`? If this would be the case, `a` and `b` would share the same container, and every access to `a` after line 9 would yield `[9,2,3,4]`. If `a` in fact is a place holder, it will always denote the array `[1,2,3,4]`, no matter what functions have obtained `a` as an argument.

To answer this question, you may simply shift the first call of `print` two lines down. As you can see, in SaC, variables are indeed place holders.

**A note for efficiency freaks:**

You may wonder whether this implies that `modarray` always copies the entire array. In fact, it only copies `a` if the place-holder property would be violated otherwise.

As a result of this place-holder property, it is guaranteed that no function call can affect the value of its arguments. In other words, the underlying concept
CHAPTER 2. ARRAY PROGRAMMING BASICS

guarantees, that all functions are "pure". Although this helps in avoiding nasty errors due to non-intended side-effects, it sometimes seems to be an annoyance to always invent new variable names, in particular, if arrays are to be modified successively.

To cope with this problem, in SaC, variables do have a so-called scope, i.e., each variable definition is associated to a well-defined portion of program code where its definition is valid. In a sequence of variable definitions, the scope of a variable starts with the left-hand side of its definition and either reaches down to the end of the function, or, provided at least one further definition of a variable with the same name exists, to the right-hand side of the next such definition. This measure allows us to reuse variable names. A slight modification of our example demonstrates the effect of variable scopes in SaC:

Listing 2.5: Variable Scopes

```sacl
use StdIO: all;
use Array: all;

int main()
{
    a = [1,2,3,4];
    b = modarray( a, [0], 9);
    print( a);
    a = b;
    print( a);
    a = modarray( a, [1], 8);
    print(a);
    a = modarray( a, [0], 0);
    print(b);
    return(0);
}
```

Here, the use of `a` on the right-hand side of line 9 still refers to the definition of line 6, whereas the use in line 11 refers to the definition in line 10.

The definition in line 13 shows, how variable scopes can be used to specify code that looks very much "imperative". However, you should always keep in mind, that in SaC, the place-holder property always holds!

Exercise 4 What result do you expect from the following SaC program?

Listing 2.6: Scope Exercise

```sacl
use StdIO: all;
use Array: all;

int main()
{
    a = [1,2,3,4];
    b = [a,a];
    a = modarray( modarray( a, [0], 0), [1], 0);
    b = modarray( b, [0], a);
    print(b);
    return(0);
}
```
2.2 Lesson: Shape-Invariant Programming

The term shape-invariant programming refers to a programming style where all array operations are defined in terms of operations that are applied to entire arrays rather than to individual array elements. In order to realize such a programming style, it is an essential prerequisite to be able to apply functions to arbitrarily shaped arguments. In SaC, this is the case.

All built-in array operations as well as all array operations supplied by the standard library can be applied to arguments of arbitrary shapes. However, most of the operations that require more than one argument do have certain restrictions with respect to which array shapes can be combined as valid arguments. If an operation is applied to a set of arguments whose shapes constitute an illegal combination, usually, a type error message is given. In cases where this cannot be decided statically, the compiler may accept such a kind of domain error and produce a runtime error instead.

This lesson consists of three parts: The section on Standard Array Operations introduces the most important standard array operations provided by the current SaC compiler release\(^1\). The next section explains Axis Control Notation, a powerful but simple way of manipulating the focus of array operations with respect to individual axes of argument arrays. With the axis-control notation, the basic operations often can easily be combined into rather complex operations as demonstrated in the section on Putting it all Together.

2.2.1 Standard Array Operations

In the sequel, several toy examples demonstrate the functionality of the most basic array operations that come as part of the current SaC release. Their design is inspired by those available in APL. However, several aspects - in particular regarding the treatment of special cases in APL - have been adjusted to allow for a more favourable compilation in SaC that yields better runtime performance.

A note for language design freaks:

You may have your own ideas on what primitive array operations should be available and how the precise semantics of these should look like. Therefore, it should be mentioned here, that all(!) array operations introduced in the remainder of this section are not hard-wired into the language, but they are defined in the module Array from the standard library. This is to say that the advanced SaC programmer may write his own set of standard array operations.

The individual parts of this section are all organized according to the following scheme: first, a semi-formal introduction to the functionality of individual operations is given. Then, several examples shed some more light on the exact semantics of each operation by varying the argument-shape constellations and

\(^1\)As of this writing, the latest SaC compiler release is version 1.00-beta.
by exploring "border-line cases" with respect to domain restrictions if these do exist.

Basic Operations

The most basic operations are very close to the model of arrays in SaC. They comprise functions for inspecting, creating, and modifying an array’s shape and content. If not stated otherwise, they are applicable to arbitrarily shaped arguments of built-in element type. Note here, that some of them have been introduced in earlier lessons already.

\( \text{dim}(a) \) returns the (scalar) dimensionality of the argument array \( a \).

\textit{Domain restrictions:} none.

\( \text{shape}(a) \) returns the shape vector of the argument array \( a \).

\textit{Domain restrictions:} none.

\( a[iv] \) constitutes a short-cut notation for \( \text{sel}(iv, a) \). It selects the array element of \( a \) at index position \( iv \). As \( a \) may be of any shape, the index position is given as an \textit{index vector}. The dimensionality of the result is identical to the dimensionality of \( a \) minus the length of \( iv \). Accordingly, its shape is derived from the last components of the shape of \( a \).

\textit{Domain restrictions:}

\( \bullet \) \( \text{dim}(iv) = 1 \)

\( \bullet \) \( \text{shape}(iv)[[0]] \leq \text{dim}(a) \)

\( \bullet \) \( \forall i \in \{0, \ldots, \text{shape}(iv)[[0]]\} : iv[[i]] < \text{shape}(a)[[i]]. \)

\( a[iv]=expr \) is a short-cut notation for an assignment of the form \( a = \text{modarray}(a, iv, expr); \). The result of this application is a new array which is almost identical to \( a \). Only the element (subarray) at index position \( iv \) is different; it is replaced by \( expr \).

\textit{Domain restrictions:}

\( \bullet \) \( \text{dim}(iv) = 1 \)

\( \bullet \) \( \text{shape}(iv)[[0]] \leq \text{dim}(a) \)

\( \bullet \) \( \forall i \in \{0, \ldots, \text{shape}(iv)[[0]]\} : iv[[i]] < \text{shape}(a)[[i]] \)

\( \bullet \) \( \text{shape}(expr) = \text{shape}(a[iv]). \)

\( \text{reshape}(shp, expr) \) computes an array with shape vector \( shp \) and data vector identical to that of \( expr \).

\textit{Domain restrictions:}

\( \bullet \) \( \text{dim}(shp) = 1 \)

\( \prod_{i=0}^{\text{shape}(shp)[[0]]-1} shp[[i]] = \prod_{i=0}^{\text{dim}(expr)-1} \text{shape}(expr)[[i]]. \)
genarray( shp, expr) generates an array of shape shp, whose elements are all identical to expr.

Domain restrictions: \(\text{dim}(\text{shp}) = 1\).

Although these operations are fairly self-explaining or known from Lesson 2.1 on Arrays as Data, let us have a look at a few example applications:

Listing 2.7: Basic Operations

```c
use StdIO: all;
use Array: all;

int main()
{
    vect = [1,2,3,4,5,6,7,8,9,10,11,12];
    mat = reshape([3,4], vect);
    print(mat);
    print(mat[[1,1]]);
    print(mat[[2]]);
    print(mat[[n]]);
    mat[[1,1]] = 0;
    print(mat);
    mat[[2]] = [0,0,0,0];
    print(mat);
    mat[[n]] = genarray([3,4], 0);
    print(mat);
    empty_vect = [];
    print(empty_vect);
    empty_mat = reshape([22,0], empty_vect);
    print(empty_mat);
    print(dim(empty_mat));
    print(shape(empty_mat));
    return(0);
}
```

The different selections in Lines 11-13 show how the dimensionality of the selected element increases as the length of the index vector decreases. If the index vector degenerates into an empty vector, the entire array is selected. Similarly, the applications of modarray in Lines 15-20 demonstrate the successive replacement of individual elements, rows, or the entire array.

Lines 22-27 are meant to draw the reader’s attention to the fact that there exists an unlimited number of distinct empty arrays in SaC!

**Exercise 5** Assuming `mat` to be defined as in the previous example, what results do you expect from the following expressions:

- `reshape([3,0,5], [])[][]`?
- `reshape([3,0,5], [])[][1]`?
- `reshape([3,0,5], [])[][0]`?
- `mat[reshape([2,0], [])]`?
Element-wise Extensions

Most of the operations that can be found as standard operations on scalars in other languages are applicable to entire arrays in SaC. Their semantics are simply element-wise extensions of the well-known scalar operations. The binary operations in general do have some domain restrictions. First, the element types of both arguments do have to be identical. Furthermore, either one of the arguments has to be a scalar value, or both arguments have to have identical shapes. In the former case, the value of the scalar argument is combined with each element of the (potentially non-scalar) other argument, which dictates the shape of the result. The latter case results in an array of the same shape as both arguments are, with the values being computed from the corresponding elements of the argument arrays.

In detail, the following operations are available:

**arithmetic operations** including addition \((e_1 + e_2)\), subtraction \((e_1 - e_2)\), negation \((- e_1)\), multiplication \((e_1 * e_2)\), and division \((e_1 / e_2)\). Furthermore, division on rational numbers \((e_1 \div e_2)\) and a modulo operation \((e_1 \% e_2)\) are supported.

*Domain restrictions:* the element types have to be numerical.

**logical operations** including conjunction \((e_1 \&\& e_2)\), disjunction \((e_1 || e_2)\), and negation \((! e_1)\).

*Domain restrictions:* the element types have to be Boolean.

**relational operations** including less-than \((e_1 < e_2)\), less-or-equal \((e_1 <= e_2)\), equal \((e_1 == e_2)\), not-equal \((e_1 != e_2)\), greater-or-equal \((e_1 >= e_2)\), and greater-than \((e_1 > e_2)\).

*Domain restrictions:* none.

\(\text{max}(e_1, e_2), \text{min}(e_1, e_2)\) compute the element-wise maximum and minimum, respectively.

*Domain restrictions:* none.

**where** \((p, e_1, e_2)\) is an element-wise extension of a conditional. It expects \(p\), \(e_1\), and \(e_2\) either to have identical shapes or to be scalar. If at least one of the three arrays is non-scalar, that shape serves as shape of the result, whose values are taken from \(e_1\) or \(e_2\) depending on the value(s) of \(p\).

*Domain restrictions:*

- the element type of \(p\) has to be boolean
- the element types of \(e_1\) and \(e_2\) have to be identical
- \(\exists \text{shp} : (\text{shape}(p) = \text{shp} \lor \text{shape}(p) = []) \land (\text{shape}(e_1) = \text{shp} \lor \text{shape}(e_1) = []) \land (\text{shape}(e_2) = \text{shp} \lor \text{shape}(e_2) = [])\).

Again, these operations are fairly self-explanatory. Nevertheless, we present a few examples:
Listing 2.8: Elementwise Extensions

```plaintext
use StdIO: all;
use Array: all;

int main()
{
  vect = [1,2,3,4,5,6,7,8,9];
  mat = [ vect, vect+10, vect+20];
  print( mat);

  mat2 = where( (mat % 2) == 0, mat, -mat);
  print(mat2);

  print( max( mat2, 0));
  return(0);
}
```

The most interesting part of this example is the definition of the matrix `mat2` in Line 11. The even numbers from the matrix `mat` are taken as they are, whereas the odd numbers are negated. Note here, that all sub expressions in predicate position are in fact non-scalar arrays: `(mat % 2)` denotes a matrix of zeros and ones and `(mat % 2) == 0` denotes a matrix of boolean values.

**Exercise 6** What results do you expect from the following expressions:

- `min( reshape([3,0,5], []), 42)`?
- `reshape([3,0,5], []) + reshape([3,0,5], [])`?
- `reshape([1,1], [1]) + reshape([1], [1])`?

**Restructuring Operations**

The operations to be introduced here do not compute new values at all. Instead, they are meant to create slightly differently structured arrays from existing ones. Therefore, they are applicable to arrays of all built-in element types.

`take(sv, a)` takes as many elements from the array `a` as indicated by the shape vector `sv`. Each element of `sv` corresponds to one axis of `a` starting from the leftmost one. For positive components of `sv`, the elements are taken from the "beginning", i.e., starting with index 0, otherwise they are taken from the "end" including the maximum legal index of the corresponding axis. All axes of `a` where there exists no corresponding element in `sv` are taken entirely.

**Domain restrictions:**

- `dim(sv) = 1`
- `shape(sv)[[0]] ≤ dim(a)`
- `∀i ∈ {0,...,shape(sv)[[0]]}: |sv[[i]]| ≤ shape(a)[[i]]`
\textbf{drop( sv, a)} drops as many elements from the array \textit{a} as indicated by the shape vector \textit{sv}. Each element of \textit{sv} corresponds to one axis of \textit{a} starting from the leftmost one. For positive components of \textit{sv}, the elements are dropped from the "beginning", i.e., starting with index 0, otherwise they are dropped from the "end" starting from the maximum legal index of the corresponding axis. All axes of \textit{a} where there exists no corresponding element in \textit{sv} are left untouched. \\
\textit{Domain restrictions:} \\
\quad \bullet \ dim(sv) = 1 \\
\quad \bullet \ shape(sv)[[0]] \leq \dim(a) \\
\quad \bullet \ \forall i \in \{0, \ldots, \text{shape}(sv)[[0]]\}: \ |sv[[i]]| \leq \text{shape}(a)[[i]] \\

\textbf{tile( sv, ov, a)} takes a tile of shape \textit{sv} from \textit{a} starting at the index specified by the offset vector \textit{ov}. For axes where no values of \textit{sv} or \textit{ov} are specified these are assumed to be identical to the extent of \textit{a} along that axis or 0, respectively. \\
\textit{Domain restrictions:} \\
\quad \bullet \ dim(sv) = \dim(ov) = 1 \\
\quad \bullet \ shape(sv)[[0]] \leq \dim(a) \\
\quad \bullet \ shape(ov)[[0]] \leq \dim(a) \\
\quad \bullet \ \forall i \in \{0, \ldots, \text{shape}(ov)[[0]]\}: \ ov[[i]] \leq \text{shape}(a)[[i]] \\
\quad \bullet \ \forall i \in \{0, \ldots, \min(\text{shape}(ov)[[0]], \text{shape}(sv)[[0]])\}: \ ov[[i]] + sv[[i]] \leq \text{shape}(a)[[i]] \\
\quad \bullet \ \forall i \in \{\min(\text{shape}(ov)[[0]], \text{shape}(sv)[[0]]), \ldots, \text{shape}(sv)[[0]]\}: \ sv[[i]] \leq \text{shape}(a)[[i]] \\

\textbf{e1 ++ e2} concatenates arrays \textit{e1} and \textit{e2} with respect to the leftmost axis. As in SaC all arrays are homogeneous, this requires all but the leftmost axis to be of identical extent. \\
\textit{Domain restrictions:} \\
\quad \bullet \ \textit{e1} and \textit{e2} have to be of identical element type \\
\quad \bullet \ \text{drop( 1, shape(e1))} = \text{drop( 1, shape(e2))}. \\

\textbf{rotate( ov, a)} rotates the array \textit{a} with respect to those axes specified by the offset vector \textit{ov}. Starting from the leftmost axis, the elements of \textit{ov} specify by how many positions the elements are rotated towards increasing indices (positive values) or towards decreasing indices (negative values). \\
\textit{Domain restrictions:} \\
\quad \bullet \ \dim(ov) = 1 \\
\quad \bullet \ shape(ov)[[0]] \leq \dim(a)
**shift( ov, expr, a)** shifts the array a with respect to those axes specified by the offset vector ov. The element positions that become "void" are filled by the (scalar) default element expr. Again, depending on the sign of the values of ov the elements are either shifted towards increasing or decreasing indices.

**Domain restrictions:**
- $\dim(ov) = 1$
- $\shape(ov)[[0]] \leq \dim(a)$
- $\shape(expr)[[0]] = []$

A few examples:

```
Listing 2.9: Restructuring Operations

```

```c
use StdIO: all;
use Array: all;

int main()
{
    vect = [1,2,3,4,5,6,7,8,9];
    mat = [ vect, vect+10, vect+20];
    print( mat);
    print( take( [2,-2], mat ) );
    print( take( [2], mat ) );
    print( take( [], mat ) );
    print( take( [0], mat ) );
    print( take( [2], reshape( [3,0,5], [] ) ) );
    print( drop( [0,-1], mat ) );
    print( mat ++ mat);
    print( rotate( [-1,42], mat ) );
    print( rotate( [1], mat ) );
    print( shift( [0,-2], 0, mat ) );
    print( shift( [0,-22], 0, mat ) );
    print( shift( [1], 0, mat ) );
    return(0);
}
```

The applications of \texttt{take} in Lines 11-13 demonstrate, how the dimensionality of \texttt{mat} remains unaffected by the length of the first argument. Only the shape of the result and the "side" from which the elements are taken is defined by it.

The applications in Lines 15-17 demonstrate how empty arrays are dealt with in the individual argument positions. In particular from the example in Line 17 it can be seen how well the concept of having an unlimited number of different empty arrays available fits nicely into the overall framework.

The remaining examples are rather straightforward. The only aspect of interest here may be the "overflows" in the rotation and shift parameters in Lines 24 and 28, respectively.
Exercise 7  Which of the following expressions can be reformulated in terms of \texttt{take}, \texttt{++}, and the basic operations defined in the previous parts?

- \texttt{drop( v, a)}?
- \texttt{tile( v, o, a)}?
- \texttt{shift( [n], e, a)}?
- \texttt{shift( [m,n], e, a)}?
- \texttt{rotate( [n], a)}?
- \texttt{rotate( [m,n], a)}?

Can we define the general versions of \texttt{shift} and \texttt{rotate} as well?

Reduction Operations

The library of standard array operations that comes with the current SaC release also contains a set of functions that fold all (scalar) elements of an array into a single one. The most common ones of these are described here.

\texttt{sum( a)} sums up all elements of the array \texttt{a}. If \texttt{a} is an empty array, 0 is returned.
\hspace{1cm} \textit{Domain restrictions:} the element type has to be numerical.

\texttt{prod( a)} multiplies all elements of the array \texttt{a}. If \texttt{a} is an empty array, 1 is returned.
\hspace{1cm} \textit{Domain restrictions:} the element type has to be numerical.

\texttt{all( a)} yields \texttt{true}, iff all elements of \texttt{a} are \texttt{true}. If \texttt{a} is an empty array, \texttt{true} is returned.
\hspace{1cm} \textit{Domain restrictions:} the element type has to be boolean.

\texttt{any( a)} yields \texttt{true}, iff at least one element of \texttt{a} is \texttt{true}. If \texttt{a} is an empty array, \texttt{false} is returned.
\hspace{1cm} \textit{Domain restrictions:} the element type has to be boolean.

\texttt{maxval( a)} computes the maximum value of \texttt{a}. If \texttt{a} is an empty array, the minimal number of the according element type is returned.
\hspace{1cm} \textit{Domain restrictions:} the element type has to be numerical.

\texttt{minval( a)} computes the minimum value of \texttt{a}. If \texttt{a} is an empty array, the maximal number of the according element type is returned.
\hspace{1cm} \textit{Domain restrictions:} the element type has to be numerical.

A few examples:
Listing 2.10: Reduction Operations

```c
use StdIO: all;
use Array: all;

int main()
{
  vect = [1,2,3,4,5,6,7,8,9];
  mat = [ vect, vect+10, vect+20];

  print( sum( mat) );
  print( prod( vect) );
  print( all( mat >= 1) );
  print( any( mat > 1) );
  print( maxval( mat) );
  print( minval( mat) );

  return(0);
}
```

Most of these examples, again, are fairly self explanatory. However, you may get an idea of the specificational advantages of shape-invariant programming when having a closer look at Lines 12 and 13. They demonstrate the rather intuitive style of program specifications that results from it.

**Exercise 8** All operations introduced in this part apply to all elements of the array they are applied to. Given the array operations introduced so far, can you specify row-wise or column-wise summations for matrices? Try to specify these operations for a 2 by 3 matrix first.

### 2.2.2 Axis Control Notation

As can be seen from Exercise 8, without further language support, it is rather difficult to apply an array operation to certain axes of an array only. This section introduces two language constructs of SaC which, when taken together, can be used to that effect. While **Generalized Selections** are convenient for separating individual axes of an array, **Set Notations** allow to recombine such axes into a result array after applying arbitrary operations to them. However, as the two constructs in principle are orthogonal, we introduce them separately before showing how they can be combined into an instrument for **Axis Control**.

**Generalized Selections**

The selection operation introduced in Section 2.2.1 does not only allow scalar elements but entire subarrays of an array to be selected. However, the selection of (non-scalar) subarrays always assumes the given indices to refer to the leftmost axes, i.e., all elements with respect to the rightmost axes are actually selected. So far, a selection of arbitrary axes is not possible. As an example use-case consider the selection of rows and columns of a matrix. While the former can be done easily, the latter requires the array to be transposed first.

To avoid clumsy notations, SaC provides special syntactical support for selecting arbitrary subarrays called **Generalized Selections**. The basic idea is to
indicate the axes whose elements are to be selected entirely by using dot-symbols instead of numerical values within the index vectors of a selection operation.

Note here, that vectors containing dot-symbols are not first class citizens of the language, i.e., they can exclusively be specified within selection operations directly!

There are two kinds of dot-symbols, single-dots which refer to a single axis and triple-dots which refer to as many axes as they are left unspecified within a selection. In order to avoid ambiguities, a maximum of one triple-dot symbol per selection expression is allowed.

A few examples:

```plaintext
Listing 2.11: Generalized Selections

use StdIO: all;
use Array: all;

int main()
{
    vect = [1,2,3,4,5,6,7,8,9];
    mat = [ vect, vect+10, vect+20];
    print( mat);

    print( mat[[1]]);
    print( mat[[1,1]]);
    print( mat[[1,..]]);

    print( mat[[..1]]);
    print( mat[[..,1]]);
    print( mat[[1,..,1]]);

    arr3d = [ mat, mat];
    print( arr3d);

    print( arr3d[[..,1]]);
    print( arr3d[[..,1,1]]);

    return(0);
}
```

The examples in Lines 11-13 demonstrate different versions for selecting the second row of the matrix mat. However, as the rightmost axis is to be selected, a dot-free version (cf. Line 11) suffices for this task. The selection of the second column of mat is shown in Lines 15 and 16.

Line 18 demonstrates that the triple-dot notation can also be successfully applied if no axis can be matched at all.

The difference between the single-dot and the triple-dot notation is shown in Lines 23 and 24. While the selection in Line 23 is identical to arr3d[[..,1,1]], the one in Line 24 is identical to arr3d[[..,1]].

Only in cases where the number of single-dots plus the number of numerical indices exceeds the number of axes available, an error message will be generated.

**Exercise 9** How can a selection of all elements of mat be specified using generalized selections? Try to find all 9 possible solutions!
Exercise 10  Referring to Exercise 5, can this new notation be used for selecting "over" empty axis? For example, can you specify a selection vector $\langle \text{vec} \rangle$, so that \( \text{reshape}([3,0,5], [])[\langle \text{vec} \rangle] = \text{reshape}([3,0], []) \) holds?

Set Notation

The means for composing arrays that have been described so far are rather restricted. Apart from element-wise definitions all other operations treat all elements uniformly. As a consequence, it is difficult to define arrays whose elements differ depending on their position within the array. The so-called set notation facilitates such position dependent array definitions. Essentially, it consists of a mapping from index vectors to elements, taking the general form

\[
\{ \langle \text{index\_vector} \rangle \rightarrow \langle \text{expression} \rangle \}
\]

where $\langle \text{index\_vector} \rangle$ either is a variable or a vector of variables and $\langle \text{expression} \rangle$ is a SAC expression that refers to the index vector or its components and defines the individual array elements. The range of indices this mapping operation is applied to usually can be determined by the expression given and, thus, it is not specified explicitly.

A note for language design freaks:

You may wonder why we restrict the expressiveness of the set notation by relying on compiler driven range detection rather than an explicit range specification. The reason for this decision is the observation that in many situations the capabilities of the set notation suffice whereas an explicit specification of ranges would obfuscate the code. Furthermore, as you will see in chapter 4, SAC provides a more versatile language construct for defining arrays. However, the expressiveness of that construct comes for quite some specification overhead.

Let us have a look at some examples:

Listing 2.12: Basic Set Notation

```plaintext
use StdIO: all;
use Array: all;

int main()
{
    vect = [0,1,2,3,4,5,6,7,8,9];
    mat = { [i] -> vect[[i]]*10+vect };
    print( mat);
    mat_inc = { iv -> mat[iv] + 1};
    print( mat_inc);
    mat_trans = { [i,j] -> mat[[j,i]] };
    print( mat_trans);
    mat_diag = { [i,j] -> where( i==j , mat[[i,j]] , 0) };
    print( mat_diag);
    return(0);
}
```
CHAPTER 2. ARRAY PROGRAMMING BASICS

The set notation in Line 8 defines a vector whose components at position \([i]\) are vectors that are computed from adding a multiple of 10 to the vector \(\text{vect}\). The legal range of \(i\) is derived from the selection \(\text{vect}[\[i\]]\) yielding in fact a matrix with shape \([10,10]\). An explicit element-wise increment operation is specified in Line 11. Since the operation does not need to refer to individual axes a variable \(iv\) is used for the entire index vector rather than having variables for individual index components. Line 14 shows how the matrix can be transposed, and Line 17 changes all non-diagonal elements to 0.

Exercise 11 Which of these operations can be expressed in terms of the array operations defined so far?

Exercise 12 What results do you expect if \(\text{mat}\) is an empty matrix, e.g., \(\text{reshape( [10,0], [])}\)?

As we can see from the set notation in Line 8, non-scalar expressions within the set notation per default constitute the inner axes of the result array. This can be changed by using .-symbols for indicating those axes that should constitute the result axis.

A few examples:

Listing 2.13: Advanced Set Notation

```plaintext
use StdIO: all;
use Array: all;

int main()
{
    vect = [0,1,2,3];
    mat = { [.,i] -> vect[i]*4 + vect }; 
    print( mat);

    arr3d = { [i] -> vect[i]*16 + mat};
    print( arr3d);

    arr3d = { [.,.,i] -> vect[i]*16 + mat};
    print( arr3d);

    arr3d = { [.,i] -> vect[i]*16 + mat};
    print( arr3d);

    return(0);
}
```

These examples show how the result of evaluating the expression on the right of the arrow can be directed into any axes of the overall result array. As can be seen in Line 17, the axes of the expressions can even be put into non-adjacent axes of the result.

Exercise 13 The .-symbol in the set notation allows us to direct a computation to any axes of the result. This is identical to first putting the result into the innermost axes and then transposing the result. Can you come up with a general scheme that translates set notations containing .-symbols into set notations that do without?
Axis Control

Although generalized selections and the set notation per se can be useful, their real potential shows when they are used in combination. Together, they constitute means to control the axes a given operation is applied to.

The basic idea is to use generalized selections to extract the axes of interest, apply the desired operation to the extracted subarrays and then recombine the results to the overall array.

For example, we can now easily sum up the individual rows or columns of a matrix:

Listing 2.14: Axis Control: sum

```plaintext
use StdIO: all;
use Array: all;

int main()
{
    vect = [0,1,2,3,4,5,6,7,8,9];
    mat = {[.,i] -> vect[[i]]*10+vect};
    print( mat);
    sum_rows = {[i] -> sum( mat[[i]] )};
    print( sum_rows);
    sum_cols = {[i] -> sum( mat[[.,i]] )};
    print( sum_cols);
    return(0);
}
```

Reduction operations, in general, are prone to axis control, as they often need to be applied to one or several particular axes rather than an entire array. Other popular examples are the maximum (maxval) and minimum (minval) operations:

Listing 2.15: Axis Control: max

```plaintext
use StdIO: all;
use Array: all;

int main()
{
    vect = [0,1,2,3];
    arr3d = {[i,j] -> vect[[i]]*4 + vect[[j]]*16 + vect};
    print( arr3d);
    max_inner_vects = {[i,j] -> maxval( arr3d[[i,j]] )};
    print( max_inner_vects);
    max_inner_arrays = {[i] -> maxval( arr3d[[.,i]] )};
    print( max_inner_arrays);
    max_outer_arrays = {[i] -> maxval( arr3d[[.,.,i]] )};
    print( max_outer_arrays);
    return(0);
}
```
In Line 8, we directly generate a 3-dimensional array from the vector `vect`. Lines 11, 14, and 17 compute maxima within different slices of that array. `max_inner_vecs` is a matrix containing the maxima within the innermost vectors, i.e., the 3-dimensional array is considered a matrix of vectors whose maximum values are computed. For `max_inner_arrays`, the array is considered a vector of matrices; it contains the maximum values of these subarrays. The last example demonstrates, that outer dimensions can be considered for reduction as well.

Further demand for axis control arises in the context of array operations that are dedicated to one fixed axis (usually the outermost one) and that need to be applied to another one. Examples for this situation are the concatenation operation (++) and reverse:

Listing 2.16: Axis Control: ++, reverse

```plaintext
use StdIO: all;
use Array: all;

int main()
{
  vect = [0,1,2,3];
  arr3d = { [i,j] -> vect[i]*4 + vect[j]*16 + vect }; print( arr3d );
  print( arr3d ++ arr3d );
  print( { [i] -> arr3d[i] ++ arr3d[i] } );
  print( { [i,j] -> arr3d[i,j] ++ arr3d[i,j] } );
  print( reverse( arr3d ) );
  print( { [i] -> reverse( arr3d[i] ) } );
  print( { [i,j] -> reverse( arr3d[i,j] ) } );
  return(0);
}
```

Line 11 shows a standard application of the concatenation of two arrays. It affects the outermost axis only, resulting in an array of shape [8, 0, 0]. The two subsequent lines show, how to apply concatenation to other axis. Essentially, the selections on the right hand sides select the sub expressions to be concatenated and the surrounding set notation glues the concatenated subarrays back together again.

The examples in Lines 15-17 show the same exercise for the operation `reverse` which reverses the order of the elements within an array with respect to the outermost axis.

**Exercise 14** The operation `take` is defined in a way that ensures inner axes to be taken completely in case the take vector does not provide enough entities for all axes. How can `take` be applied to an array so that the outermost axis remains untouched and the selections are applied to inner axes, starting at the second one? (You may assume, that the take vector has fewer elements than the
array axes!) Can you specify a term that - according to a take vector of length 1 - takes from the innermost axis only?

**Exercise 15** Can you merge two vectors of identical length element-wise? Extend your solution in a way that permits merging n-dimensional arrays on the outermost axis.

### 2.2.3 Putting it all Together

The array operations presented so far constitute a substantial subset of the functionality that is provided by array programming languages such as APL. When orchestrated properly, these suffice to express rather complex array operations very concisely. In the sequel, we present two examples that make use of this combined expressive power: matrix product and relaxation.

#### Matrix Product

The matrix product of two matrices $A$ and $B$ (denoted by $A \circ B$) is defined as follows:

Provided $A$ has as many columns as $B$ has rows, the result of $A \circ B$ has as many rows as $A$ and as many columns as $B$. Each element $(A \circ B)_{i,j}$ is defined as the scalar product of the $i$-th row of $A$ and the $j$-th column of $B$, i.e., we have \((A \circ B)_{i,j} = \sum_k A_{i,k} \times B_{k,j}\).

This definition can directly be translated into the following SaC code:

```sac
use StdIO: all;
use Array: all;

int main()
{
    id = [ [1d, 0d, 0d], [0d, 1d, 0d], [0d, 0d, 1d] ];
    vect = [1d, 2d, 3d, 4d];
    mat = [ vect, vect+4d, vect+8d ];
    print( mat );
    res = { [i,j] -> sum( id[[i,.]] * mat[[.,j]]) };
    print( res );
    return(0);
}
```

After defining two matrices `id` and `mat` in Lines 6 and 8, respectively, the matrix product `id \circ mat` is specified in Line 12. `id[[i, .]]` selects the $i$-th row of `id` and `mat[[., j]]` refers to the $j$-th column of `mat`. The index ranges for $i$ and $j$ are deduced from the accesses into `id` and `mat`, respectively. A variable $k$ as used in the mathematical specification is not required as we can make use of the array operations `*` and `sum`. 
Relaxation

Numerical approximations to the solution of partial differential equations are often made by applying so-called relaxation methods. These require large arrays to be iteratively modified by so-called stencil operations until a certain convergence criterion is met. Fig. 2.1 illustrates such a stencil operation.

A stencil operation re-computes all elements of an array by computing a weighted sum of all neighbor elements. The weights that are used solely depend on the positions relative to the element to be computed rather than the position in the result array. Therefore, we can conveniently specify these weights by a single matrix of weights as shown on the left side of Fig. 2.1.

In this example, only 4 direct neighbor elements and the old value itself are taken into account for computing a new value. (Hence its name: 5-point-stencil operation). As can be seen from the weights, a new value is computed from old ones by adding an eight-th each of the values of the upper, lower, left, and right neighbors to half of the old value.

As demonstrated on the right side of Fig. 2.1 our example assumes so-called cyclic boundary conditions. This means that the missing neighbor elements at the boundaries of the matrix are taken from the opposite sides as indicated by the elliptic curves.

In the sequel, we concentrate on the specification of a single relaxation step, i.e., on one re-computation of the entire array. This can be specified as a single line of SAC code:

```
Listing 2.18: Relaxation with Cyclic Boundaries

use StdIO: all;
use Array: all;

int main(){
    weights = [ [0d, 1d, 0d], [1d, 4d, 1d], [0d, 1d, 0d] ] / 8d;
    vect = [1d, 2d, 3d, 4d];
    mat = [ vect, vect+4d, vect+8d, vect+12d];
    print( mat);
}```
Line 6 defines the array of weights as given on the left side of Fig. 2.1. Our example array is initialized in Lines 8-9. The relaxation step is specified in Line 12. At its core, all elements are re-computed by operations on the entire array rather than individual elements. This is achieved by applying \texttt{rotate} for each legal index position \texttt{iv} into the array of weights \texttt{weights}. Since the expression \[
\{ \texttt{iv} \rightarrow \texttt{weights}[\texttt{iv}] \ast \texttt{rotate( iv-1, mat)} \}[[...,i,j]]
\] computes a 3 by 3 array of matrices (!) the reduction operation \texttt{sum} needs to be directed towards the outer two axes of that expression only. This is achieved through axis control using a selection index 
\[
[...\ldots, \ldots, i, j]]
\] within a set notation over \texttt{i} and \texttt{j}.

\textbf{Exercise 16} Another variant of relaxation problems requires the boundary elements to have a fixed value. Can you modify the above solution in a way that causes all boundary elements to be 0? \textbf{[Hint:} You may consider the boundary elements to actually be located \textbf{outside} the matrix\textbf{]}
Chapter 3

Basic Program Structure

This trail gives a brief introduction into the main language constructs most of which have been adopted from standard C. We assume some familiarity with standard C and, therefore, only give a quick overview and highlight the differences between SaC and C.

3.1 Lesson: Functions and their Types

3.1.1 Function Definitions

Like in other modern programming languages functions constitute the main form of structuring programs in SaC. SaC functions very much resemble their C counterparts. The most prominent difference is that SaC functions can have multiple return values, as illustrated in the following example.

```
use ScalarArith: all;
int, int divmod( int x, int y)
{
    return( x/y, x%y);
}
int main()
{
    d, m = divmod( 8, 3);
    return( d);
}
```

A function with multiple return values, like `divmod` in the above example, has a comma-separated list of return types in front of the function name and the `return`-statement likewise contains a comma-separated list of expressions. Obviously both lists must coincide in length to make up a well-formed program. Functions with multiple return values cannot appear in expression positions in
SaC, but their results need to be directly assigned to identifiers as illustrated above.

**Exercise 17** Extend the above example program to compute the greatest common denominator of two numbers using Euclid’s algorithm. In particular, use the function \texttt{divmod}.

### 3.1.2 Built-in Types and Subtyping

SaC supports all basic types of standard C, i.e. the integer types \texttt{short int}, \texttt{int}, \texttt{long int} and \texttt{long long int} both in signed and unsigned versions, the floating point types \texttt{float}, \texttt{double}, \texttt{long double} and \texttt{quad} (if supported on the execution platform) and the character type \texttt{char}. In addition, SaC supports three more basic types: the boolean type \texttt{bool} and the integer type \texttt{byte}, both signed and unsigned. The introduction of a specific byte type is motivated by our desire to separate numerical types from non-numerical types such as \texttt{char} and \texttt{bool}. All basic types are mapped one-to-one to their C counterparts and, hence, the same rules apply to them in SaC as in C. For example, the concrete bit widths used for representation and, hence, the range of values for integer types are platform-dependent.

For each basic type there is an entire hierarchy of array types that specify the shape of an array (remember that any expression in SaC denotes an array) at different levels of accurateness. Using type \texttt{int} as a running example, \texttt{int} itself denotes an integer array with rank zero, the empty vector \[ \] as shape and a single element, in other words the equivalent of a scalar value. Then, there are (real) array types denoting arrays of rank greater than zero, e.g. \texttt{int[4]} denotes a 4-element vector while \texttt{int[10,20]} denotes a 10 by 20 element matrix.

Whereas all these types specify exact shapes, SaC also features types that solely denote the rank of some array, but leave the concrete shape open, e.g. \texttt{int[.]} describes the type of all integer vectors (of any length) and \texttt{int[.,.]} is the type of all integer matrices. In order to support rank-invariant programming, SaC, furthermore, has types that not only abstract from concrete shapes but even from concrete ranks. These are \texttt{int[*]} which is the type of all integer arrays of any rank and shape, including rank-zero arrays (usually referred to as scalars) and \texttt{int[+]}, the type of all “true” integer arrays, i.e. arrays of rank greater than zero.

SaC defines a subtype relationship between array types in the obvious way. For example, \texttt{int[*]} is a subtype of \texttt{int[*], int[.]} and \texttt{int[.,.]} are both subtypes of \texttt{int[*]} and \texttt{int[12]} and \texttt{int[42]} are subtypes of \texttt{int[.]} The subtyping hierarchy of SaC has exactly four levels.
CHAPTER 3. BASIC PROGRAM STRUCTURE

3.2 Lesson: Function Bodies

3.2.1 Variable Declarations

Local variables within bodies of SaC functions are typically not declared (in contrast to C), but the SaC compiler infers proper types for local variables or yields an appropriate error message. Nevertheless, it is syntactically legal to add explicit variable declarations in SaC in exactly the same way as in C.

There is one difference to C, however: While C allows local variable declarations at the beginning of each code block and in the latest C99 standard instructions and declarations can even be interleaved, SaC only supports variable declarations on the level of function bodies, and they must precede any instruction.

**Exercise 18** Rewrite your solution to computing the greatest common denominator of two numbers from the previous exercise such that each subexpression is assigned to an identifier, i.e. flatten any nested expression. Then add explicit variable declarations for each local variable used.

**Exercise 19** The previous exercise only used scalar types, more precisely `int`. What happens if you replace `int` in all variable declarations by its supertype `int[*]`?

3.2.2 Assignments

As we have already seen in the previous trails, SaC allows for C-style assignments to variables. In contrast to C, assignments cannot be placed within expression positions and the comma-operator of C is not supported. However, the combinations of operators and assignment are the same in SaC as in C.

```plaintext
use ScalarArith: all;
use StdIo: all;

int main()
{
    a = 42;
    print( a);
    a += 5;
    print( a);
    a -= a;
    print( a);
    return( a);
}
```

**Listing 3.2: Operator assignments in SaC**

Note that despite the term “Single Assignment” in the name of SaC, the language actually supports repeated assignment of values to the same variable as in the above example. This seeming contradiction can be explained as follows: Each assignment opens up a new scope of an identifier bound to some value. Accordingly, the second assignment to `a` in the above example opens up a new
scope for a new identifier \( a \) that only coincidentally carries the same name as
the identifier introduced in the code line before. However, because these two
variables do carry the same name, the second assignment shadows the scope of
the first assignment meaning that no access to the first \( a \) is possible any more.

### 3.2.3 Conditionals

In SaC, we support three forms of conditionals:

- branching with consequence only (if-then),
- branching with consequence and alternative (if-then-else) and
- conditional expressions.

All three forms use a syntax that is identical to that of C, as the following
listing illustrates. Similarity with C extends to the use of curly brackets to
build blocks of multiple statements and their potential absence if the condition
covers a single statement only.

**Listing 3.3: Conditionals in SaC**

```c
use ScalarArith: all;
use StdIO: all;

int main( )
{
    a = 5;
    b = 7;
    printf( "a=%d, b=%d\n", a, b);
    if (a<b) a = b;
    printf( "a=%d, b=%d\n", a, b);
    if (a>=b) {
        a = b;
    } else {
        b = a;
    }
    printf( "a=%d, b=%d\n", a, b);
    b = a<b ? a : b;
    printf( "a=%d, b=%d\n", a, b);
    return( b);
}
```

As in C all forms of conditionals can be nested in any way, and the dangling
else ambiguity is resolved as in C proper.

A small difference to standard C is that the predicate expression of any
conditional must be of type `bool`. There is no implicit treatment of integer
values as predicates. Another subtle difference to C stems from the functional
nature of SaC: a variable defined only in one branch of a conditional, will cause
the SaC compiler to raise an error because the value may be undefined.
The switch-statement of C is currently not supported by SaC. This is not so much motivated by conceptual concerns, but rather by pragmatic considerations like the ratio between expressiveness gained and implementation effort caused.

### 3.2.4 Loops

SaC supports all three loop constructs of standard C: while, do and for with the familiar syntax, as illustrated by the following code fragment.

**Listing 3.4: Loops in SaC**

```c
use ScalarArith: all;
use StdIO: all;

int main()
{
    a = 10;
    while (a>0) {
        a = a - 2;
        print(a);
    }
    do {
        print(a);
        a = a + 1;
    } while (a<7);
    for (i=1, j=2 ; i+j < 42 ; i++, j++) {
        a *= 2;
        print(a);
    }
    return(a);
}
```

In analogy to conditionals, the loop predicate expression must be of type bool. Note that SaC does even support the comma operator in for-loops (though not in general terms as pointed out before).

These C-style loop constructs can make code look very imperative. Despite these syntactic similarities, always bear in mind that SaC loops are (only) syntactic sugar for equivalent tail-end recursive functions. While the functional semantics almost completely coincides with the pragmatic expectations of a C programmer, some subtle issues may arise concerning the definedness of variables. For example, the SaC compiler would complain about the above example saying that the variable \( b \) in the while-loop may be used uninitialised if it is not defined before. This is because the compiler assumes that the body of a while-loop may not be executed at all. Of course, you may know better, but the SaC compiler at the moment makes no particular effort to prove this fact when it analyses the definedness of variables.
CHAPTER 3.  BASIC PROGRAM STRUCTURE

3.2.5 Explicit Control Flow Manipulation

The control flow manipulation statements of C, i.e. goto, break and continue, as well as labels are not supported by SaC. This is due to the fact that SaC is indeed a functional language and as such there is actually no control flow, even though the C-like syntax suggests one.

3.3 Lesson: Advanced Topics

3.3.1 User-defined Types

SaC allows programmers to define their own types using a syntax that is identical to C.

Listing 3.5: User-defined types

```plaintext
use StdIO: all;
typedef int myint;
typedef float[100,100] real_matrix;
typedef double[2] complex;

int main()
{
    complex c;
double[2] d;
c = (:complex) [1,2,3];
d = (:double[2]) c;
print(d);
return(0);
}
```

Following the keyword typedef we have the defining type followed by the defined type name. Note that in contrast to C, defining type and defined type are not considered synonyms. Types like double[2] and complex are distinguished properly and a function that expects a value of type double[2] as an argument will not accept a value of type complex instead.

SaC requires explicit type casts to change the type of a value from the defining type to the defined type or vice versa as in the following example.

Note that for the time being any defining type in a type definition must exactly specify some shape; the various less specific types are not supported. While this looks incomplete at first glance, it is noteworthy that such type definitions would immediately lead to arrays of non-uniform shape, e.g. a matrix whose rows have different length. There is no doubt that this would be an extremely powerful extension to the homogeneously shaped arrays that SaC supports today, but it would likewise require a non-trivial extension of the code generator and runtime system that we leave for future research.
3.3.2 Type Conversions

SaC uses C-like cast expressions to change the type of an expression whenever the data representation remains unaffected, i.e. between user-defined types and their defining types or vice versa. In contrast to C, SaC does not use cast expressions to actually change data representations, e.g. when converting from an integer type to a floating point type, between floating point types of different precision or between integer types of different bit width. For all these purposes SaC uses dedicated conversion functions to express the fact that such conversions actually require an operation performed at runtime rather than just changing the type interpretation of a value.

These conversion functions are named `tobool` and `tochar` for converting into non-numerical values. For all numerical types these functions are named “to” plus an optional “u” for unsigned integer types followed by the first letter of the type name (“ll” in the case of `long long int`). The following example illustrates type conversions in SaC.

Listing 3.6: Type conversions

```c
use ScalarArith: all;
use StdIO: all;
int main()
{
    double x;
    float y;
    int z;
    x = 2.3;
    print(x);
    y = tof(x);
    print(y);
    z = toi(y);
    print(z);
    return(z);
}
```

3.3.3 Function Overloading

Another distinctive feature of SaC functions is their support for overloading. More precisely, a SaC program or module may well contain definitions of multiple functions of the same name, as long as they differ in the types or their arguments. For example, the following listing contains

Listing 3.7: Function overloading

```c
use ScalarArith: all;
use StdIO: all;
int twice(int x)
{
    return(2 * x);
}
```
double twice(double x) {
    return(2.0 * x);
}

int main() {
    a = twice(5);
    print(a);
    b = twice(5.9);
    print(b);
    return(0);
}

Function overloading in SaC can be nicely combined with the subtyping hierarchy of array types as in the following example. Here, we have three instances of the function quicksort, one for vectors of any length, one for vectors of length one and one for empty vectors. The latter two boil down to the identity function. As a result we can safely access the first element of the argument vector \( v \) in the general instance because any argument vector is guaranteed to have at least two elements.

```c
int[.] quicksort( int[.] v) 
{ 
    pivot = v[1];
    ... 
}

{ 
    return( v);
}

int[0] quicksort( int[0] v) 
{ 
    return( v);
}
```
Chapter 4

With-Loops

This trail aims at providing a hands-on introduction to the key language construct of SaC: the With-Loop. It constitutes the generalisation of the set-expression as introduced in the lesson 2.2 on Shape-Invariant Programming and can be seen as a shape-invariant form of the map-reduce template or the array comprehensions found in other functional languages. However, in contrast to these, the With-Loop was carefully designed to enable radical code optimisations as well as compilation into high-performance, concurrently executable code.

A note for language design freaks:

In fact, almost all array operations introduced in earlier trails are defined by With-Loops within the standard library. This design combines two major advantages:

- **better performance**, as the conformity enables optimisations to be more generally applicable, and
- **increased flexibility**, as the user can modify the definition of all standard operations.

The introduction of With-Loops comes in a single lesson which step-wise introduces all features and variants of With-Loops in SaC.

4.1 Lesson: With-Loop Basics

4.1.1 Basic Components

Generally, With-Loops are composed of three different components:

- sets of index vectors (referred to as *generator-ranges*),
- functions that map index vectors to arbitrary values (*generator-expressions*), and
• combining operations (with-loop operators) that take such values and construct arrays from them.

In its simplest form, a With-Loop contains one component of each kind. It then maps the function defined by the generator-expression to all index vectors from the generator-range in a data-parallel fashion. This leads to a set of index-value-pairs which are combined into a result array by means of the given with-loop operator. Let’s have a look at a simple example:

Listing 4.1: Simple With-Loop

```
use StdIO: all;
use Array: all;

int main()
{
    a = with {
        ( [0] <= iv < [5] ) : 42;
    } : genarray( [7], 0);

    print(a);
    return(0);
}
```

Here, the With-Loop in lines 6-8 computes the vector [42, 42, 42, 42, 42, 0, 0]. The generator-range is specified by the code snippet in round brackets in line 7: ([0] <= iv < [5]). It denotes the set of vectors { [0], [1], ..., [4] }. The generator-expression here is the constant 42. Hence, the mapping function \( f_{\text{map}} \) maps any index vector \( iv \) into 42, i.e., we have \( f_{\text{map}}(iv) = 42 \). Finally, the with-loop operation is specified as `genarray( [7], 0 )`. This operation computes an array of shape [7] where

\[
a[iv] = \begin{cases} 
    f_{\text{map}}(iv) & \text{if } iv \in \{ [0], [1], ..., [4] \} \\
    0 & \text{otherwise}
\end{cases}
\]

**Generator-ranges** and **generator-expressions** always occur in pairs. Jointly they form a syntactical unit, which we refer to as **generator**. As we will see later, With-Loops can contain arbitrary numbers of generators. They are enclosed in curly brackets.

**Exercise 20** What result do you expect if we eliminate the generator from the above example?

What results do you expect if we modify the generator-range in the above example into:

- \(( [-2] <= iv < [3]) \)?
- \(( [0] <= iv < [8]) \)?
- \(( [6] <= iv < [5]) \)?
- \(( [8] <= iv < [5]) \)?
- \(( [6] <= iv < [0]) \)?
CHAPTER 4. WITH-LOOPS

[Hint: You should compile these examples with the option -check c being enabled]

4.1.2 Generator Ranges

SaC offers quite some flexibility when it comes to specifying generator ranges. First of all, the use of index vectors in the bounds enables the convenient specification of $n$-dimensional index ranges. Let us look at a few examples for the 2-dimensional case:

Listing 4.2: Generator Range Specifications

```c
use StdIO: all;
use Array: all;

int main()
{
    a = with {
        ( [0,2] <= [i,j] < [5,6]) : 42;
    } : genarray( [5,6], 0);
    print(a);

    shp = [5,6];
    a = with {
        ( 0*shp <= iv < shp) : 42;
    } : genarray( shp, 0);
    print(a);

    a = with {
        ( . < iv < . ) : 42;
    } : genarray( [5,6], 0);
    print(a);

    a = with {
        ( . <= jv=[x,y] <= [4,5]) : 42;
    } : genarray([5,6], 0);
    print(a);

    a = with {
        ( . <= [i,j] <= . step [1,4] ) : 42;
    } : genarray([5,6], 0);
    print(a);

    a = with {
        ( . <= [i,j] <= . step [2,4] width [1,2] ) : 42;
    } : genarray([5,6], 0);
    print(a);

    return(0);
}
```

As we can see from the first With-Loop in lines 6-8, a vector of scalar indices can be used where we previously used the variable `iv` to denote the entire index vector. In cases where the dimensionality of the With-Loop is statically fixed, this sometimes comes in handy. However, if we want to adopt a more generic, shape-invariant programming style, it becomes mandatory to use vectors for the index variable as well as for the bounds.
CHAPTER 4. WITH-LOOPS

Note here, that the ability to use vectors rather than componentised indices and bounds is absolutely crucial here! It constitutes the enabling factor for specifying With-Loops in a shape-invariant style as the length of those vectors may remain unknown until runtime. It is this particular feature that sets the With-Loops apart from most conventional language constructs for data-parallel array operations.

The With-Loop in lines 12-14 demonstrates a typical case where the dimensionality of the resulting array is solely determined by a vector (here \texttt{shp}). A slightly more elegant way for the most frequent case is the use of a syntactical shortcut supported by SaC. The symbol "." can be used in the position for the lower and upper bound, denoting the lowest legal index and the highest legal index into the array to be created, respectively. This is exemplified in the With-Loop in lines 17-19. Note here, that this generator-range does \textbf{not} cover the entire legal index space of the resulting array! As the "." always represents legal indices, we have to make sure that we use \texttt{<=} on both sides if we want to cover the entire range. The example presented here, excludes the extreme cases and, thus, covers all inner elements of the resulting array only.

\textbf{Exercise 21} What happens if the length of the vectors within the generator-range or the shape expression in the \texttt{genarray}-operation do not match? \textbf{[Hint:} You should compile these example with the option \texttt{-check c} being enabled]\textbf{]}

The With-Loop in lines 22-24 demonstrates how both, a vector version of the index vector and scalarised versions can be made available for the generator-expression. It also exemplifies that a mix of the "."-symbol and explicit expressions can be used for the bounds.

The remaining two With-Loops demonstrate the ability to specify rectangular grids of indices. The vector that follows the keyword \texttt{step} specifies the stride of the recurrence pattern per axis. As a consequence, the With-Loop in lines 27-29 computes an array whose every fourth column is \texttt{42} starting with the very first one.

The use of the vector after the keyword \texttt{width} enables the programmer to denote more than one index per stepping period. The With-Loop in lines 32-34 computes a matrix where \texttt{1x2} blocks of the value \texttt{42} are placed in the upper left corner of each \texttt{2x4} grid of the resulting array.

\textbf{Exercise 22} Can you achieve the same result array as the last With-Loop of the above examples \textbf{without} using the \texttt{step/width} facility? \textbf{[Hint:} The solution may be surprisingly short!\textbf{]}

4.1.3 Generator Expressions

As we have seen in the previous sections, each generator expression implicitly defines a mapping function from indices to expressions. The parameters of these functions are derived from the index variables introduced in the associated generator range specifications. More complex generator expressions can be specified
by assignment blocks that can be inserted between a generator range and the associated generator expression. Here a few examples:

Listing 4.3: Non-trivial generator expressions

```plaintext
use StdIO: all;
use Array: all;

int main()
{
    a = with {
        [0,1] <= [i,j] < [6,6] {
            div, mod = divmod( i, j);
        } : mod;
    } : genarray( [6,6], 0);
    print(a);
    a = with {
        . <= [i,j] <= . ; ( i==j ? 1: 0) ;
    } : genarray( [6,6], 0);
    print(a);
    a = with {
        . <= [i] <= . {
            mval = i;
            val = sum( with {
                . <= [i] <= . ; i
            } : genarray( [mval], 0));
        } : val;
    } : genarray( [6], 0);
    print(a);
    return(0);
}
```

The first With-Loop in lines 6-10 shows a typical scenario. The function divmod returns two values rather than just one. Rather than defining an explicit mapping function that passes on the desired return value, we can specify this selection directly.

The With-Loop in lines 13-15 demonstrates how non-trivial expressions can be used even without necessitating the introduction of an assignment block.

The scope of variables that are defined in such an assignment block is strictly local to that block. Such a variable can neither be referenced within other generators of the same With-Loop nor outside of the With-Loop. Note, however, that With-loops can be arbitrarily nested. An example for such a nesting is shown in the With-Loop in lines 18-22 of the examples above.

**Exercise 23** What do you expect to happen, if a variable that is defined in such an assignment block has the same name as the index variable? Where is the "modified" version observable?

Why can the variable `mval` in the example above be safely replaced by `i`?

### 4.1.4 Reductions and further With-Loop Operations

Besides the genarray with-loop operator described so far, SaC supports a few more. These are:
• a modarray operator which "modifies" an existing array, and

• two fold operators that enable the specification of reduction operations

The modarray variant is very similar to the genarray variant. The only difference is that neither the shape of the result nor a default element for unspecified index positions are explicit. Both of these are taken from a specified array that serves as a template. The second With-Loop in lines 11-13 of the example below demonstrates this. Here, a new array \(b\) is computed from the array \(a\) by negating each second element of \(a\). As in lesson 2.1 on arrays as data, printing \(a\) in line 15 shows that in fact two different arrays have been created.

Listing 4.4: Further With-Loop operators

```plaintext
use StdIO: all;
use Array: all;

int main()
{
    a = with {
        ( . <= [i] <= .) : i;
    } : genarray( [6], 0);
    print(a);

    b = with {
    } : modarray(a);
    print(b);
    print(a);

    c = with {
        ( [0] <= iv <= [5]) : a[iv];
    } : fold( *, 1);
    print(c);

    d = with {
        ( [0] <= iv <= [5]) : a[iv];
    } : foldfix( *, 1, 0);
    print(d);
    return(0);
}
```

The two final examples demonstrate how reduction operations can be specified in SaC. The first variant (lines 17-19) is the standard variant for reductions. It requires the specification of a folding function (\(\ast\) in this case) and of a neutral element (1 here). As with the earlier variants, all index vectors from the generator range are mapped according to the generator expression. Subsequently all computed values are folded using the specified folding function. Note here, that no particular folding order is guaranteed! In order to obtain deterministic results on a multicore machine this requires the specified folding operation to be both, associative and commutative.

The final With-Loop in lines 22-24 is a slight variant of the fold version. It stems from the observation that some reductions can be shortcut when a certain fixpoint value has been met. In the given example of multiplication this is the value 0. Whether the underlying implementation makes use of this extra
information or not is not specified. The computational result of both, the fold and the foldfix variant are the same, provided that the specified fixpoint value in fact constitutes one for the specified folding operation.
Chapter 5

Case Studies

This trail will illustrate how to use SaC for real-world applications. The following exercises make use of the techniques introduced in this tutorial so far. Working through this trail shall give you some more hands-on experience with the language itself and at the same time show you how to employ SaC to solve your every-day problems in an efficient way.

5.1 Lesson: Image Processing

Digital image processing spans across a vast area of applications. Ranging from digital photography over astronomy to surgery-assisting medical imaging, most applications still share as their underlying theory some sort of basic signal processing on two-dimensional (and potentially higher-dimensional) signals. It is common to work with discretised signals, i.e., an approximation of the originally analogue, continuous version of the signal.

In this exercise we will focus on two basic image filters on static, two-dimensional, 8-bit gray-scale images with a resolution of $x_{\text{max}} \times y_{\text{max}}$, such that an image is a 2D function

$$S : \mathbb{N} \times \mathbb{N} \rightarrow \mathbb{N}$$

$$S(x, y) = \begin{cases} v_{xy}, & \text{if } x < x_{\text{max}} \wedge y < y_{\text{max}} \\ 0, & \text{otherwise} \end{cases}$$

where $\forall x, y : 0 \leq v_{xy} \leq 255$. Informally speaking, the parameters $x$ and $y$ determine the position within the image and $v_{xy}$ determines the gray-scale value of the pixel at this position.

A common technique to modify an image is to apply a filter of desired properties to it. In many cases, an image filter $F$ again is a two dimensional signal as the one described above. The application of a filter to an image is
expressed by the 2D (discrete) convolution $F \otimes S$ of these two signals:

$$F(x, y) \otimes S(x, y) = \sum_{j=-\infty}^{\infty} \sum_{k=-\infty}^{\infty} F(j, k)S(x - j, y - k)$$

With this simple form of filtering, a wealth of image modifications are expressible. Depending on the choice of the filter mask we can achieve effects such as smoothing, sharpening, edge detection, embossing and many more.

In SaC we can represent these signals as 2D arrays where $x$ and $y$ are the column and row indices and each element corresponds to one pixel. The application of a filter to an image may then be expressed using a stencil operation as it was introduced in an earlier part of this tutorial (Section 2.2): The filter mask is positioned with its center point over each pixel of the original image. At each position, the pixel value of the image and its corresponding value of the filter mask are multiplied. The pixel value at the same position in the result image is determined by the sum of these products (weighed sum).

**Exercise 24** The Sobel operator is an edge-detection filter that computes the gradient image from an input signal. The filtering process consists of three steps: In the first step, horizontal edges are detected, in the second step vertical edges are detected, and in the final step, both sub-results are combined to the resulting gradient image. The first two steps are in fact two independent operations, each of which requires its own individual filter mask:

$$F_x = \begin{pmatrix} 1 & 0 & -1 \\ 2 & 0 & -2 \\ 1 & 0 & -1 \end{pmatrix} \quad F_y = \begin{pmatrix} 1 & 2 & 1 \\ 0 & 0 & 0 \\ -1 & -2 & -1 \end{pmatrix}$$

Applying these two masks to an input image $S$ yields two sub-results, one highlighting horizontal edges ($S_x = F_x \otimes S$), the other highlighting vertical edges ($S_y = F_y \otimes S$). We compute the final gradient image by adding the two sub-results $S_{xy} = S_x + S_y$. As $S_{xy}$ may contain elements with a value greater than 255, we need to cap each pixel value at the maximum allowed value of 255. See Fig. 5.1 for an example of how an image and its gradient image may look like.

Quite the opposite to edge detection, a smoothing or blurring filter is used to make edges appear less prominent. Commonly applied filters of such kind are the $3 \times 3$ and $25 \times 25$ Gaussian blurring stencils:

$$G_9 = \frac{1}{15} \cdot \begin{pmatrix} 1 & 2 & 1 \\ 2 & 3 & 2 \\ 1 & 2 & 1 \end{pmatrix} \quad G_{25} = \frac{1}{331} \begin{pmatrix} 1 & 4 & 7 & 4 & 1 \\ 4 & 20 & 33 & 20 & 4 \\ 7 & 33 & 55 & 33 & 7 \\ 4 & 20 & 33 & 20 & 4 \\ 1 & 4 & 7 & 4 & 1 \end{pmatrix}$$

Write a program that applies these operators to a given image and outputs the result for later use:
• Write a function `apply` that takes an image and a filter mask as input and returns the convolution of the two. Use either `with-loops` or `axis-control notation` to achieve this. Take care of boundary conditions when implementing the convolution.

• Write a function `sobel` that takes an image as input and applies \( F_x \) and \( F_y \) to the image. Furthermore, add \( S_x \) and \( S_y \) with capping so that this function returns the final result of the Sobel operation.

• Write functions `gauss9` and `gauss25` that take an image as input and apply \( G_9 \) and \( G_{25} \) to the image.

• Write a `main` function that reads in an image from `stdin`, calls `sobel`, `gaussBlur9` and `gaussBlur25` on this image and then write the results back to `stdout`.

You may want to use the following skeleton for your program:

Listing 5.1: Skeleton of sobel.sac

```c
use Structures: all;
use StdIO: all;
use Clock: {time, gettime, difftime};

void writeMatrix( int[,] m)
{
  for( y=0; y<shape(m)[1]; y++) {
    for( x=0; x<shape(m)[0]; x++) {
      printf( "%d ", m[x,y]);
    }
    printf( "\n");  
  }
}

int[,] sobel( int[,] img)
{
  SY = [ [1, 2, 1], [0, 0, 0], [-1, -2, -1] ];
  /* complete this function here */
}

int[,] gaussBlur9( int[,] img)
{
  /* complete this function here */
}
```
int[,,] gaussBlur25( int[,,] img)  
{" /* complete this function here */
}  

int[*,] time timestamp( int[*] img)  
{" return( img, gettime());
}  

int main()  
{" fprintf( stderr, "\nReading image in Fibre format...\n");  
img = FibreScanIntArray( stdin);  
fprintf( stderr, "\nApplying filter ...\n");  
img, start = timestamp( img);  
img, sobel( img);  
img, end = timestamp( img);  
/* Call the other filters here */  
fprintf( stderr, "\nFinished filtering after %fs\n", difftime( end, start));  
fprintf( stderr, "\nWriting result image ...\n");  
writeMatrix( img);  
return( 0);  
}"  

Compile your source code with and without the -mt option for multi-threaded execution and experiment with various thread counts. On faster machines it might be necessary to apply the filter multiple times (copy and paste line 50 a couple of times) to see measurable speed-ups.

NB: The Fibre format encodes, in addition to the raw data, shape information. By using FibreScanIntArray this program is able process 2D images of any size and it is not fixed to statically known input sizes.

Figure 5.1: The SaC logo before (left) and after (right) edge detection
CHAPTER 5. CASE STUDIES

5.2 Lesson: Computing Mandelbrot Images

Exercise 25 This exercise is about creating a basic implementation for computing mandelbrot images.

1. To get started, you may want to use the files `mandelbrot_start.sac` and `Fractal_tier1.sac` which, in essence, contain the IO-code for visualising the mandelbrot pictures. A first running version can be obtained by implementing the missing function bodies in `Fractal_tier1_empty.sac`:
   - `escapeTime` which implements the iteration on arrays of complex numbers, and
   - `genComplexArray` which computes a two-dimensional array of complex numbers that represent a discretisation of $\mathbb{C}^2$.

2. Waiting for the final picture can be rather unpleasant if it is not clear whether the chosen fraction of $\mathbb{C}^2$ yields an interesting picture and the iteration limit is high. Therefore, as a first extension, try to modify the main function in `mandelbrot_start.sac` so that it computes the mandelbrot picture with increasing resolution without changing the overall size of the picture.

Compute resolutions $[5,5]$, $[10,10]$, ..., $[320,320]$ and display them consecutively in a $[320,320]$ display by replicating the found values accordingly.

Hint: define a function `stretchRgb` which takes an array of type `color[.,.]` and an integer stretching factor `stretch` and replicates each element of the array into a square of shape `[stretch, stretch]`.

3. The function `intArrayToMonochrome` maps all escape values into a color by means of a clut. Can you express this operation without a With-Loop?

   Hint: You may find inspiration in one of the earlier tasks!

4. Try using the compiler option `-mt` to experiment with multi-core machines!

5. Try using the compiler option `-cuda` to experiment with graphics cards!

6. Try using the compiler option `-b11:cyc` to inspect the high-level, optimised intermediate code.

Exercise 26 In this exercise, we improve the way the colours are chosen in order to obtain smoother transitions. We will use a common approach referred to as normalized iteration counts. A normalized iteration count for a point in the complex plane $z$ is computed by using the iteration count $t$ and the value at that position $z_t$ during the final iteration. Using these two values, the normalized iteration count is defined as $t_n := (t + 1) - \log_2(\log_2(|z_t|))$ for those values that escape and as 0 otherwise.
1. The module Fractal tier2.sac contains stubs for the missing functionality required in this exercise:

- `normalizedIterationCount` which implements the normalisation of iteration counts by taking the final computed value into account.
  
  Hint: The function `escapeTime` only computes the number of iterations before the value at a given position escapes. To normalize these, the final value at that position is required, as well. For this, we have provided a function `escapeValue`.

- `doubleArrayToRGB` maps the normalized iteration counts, which are double values, to an RGB colour-triple. To derive an RGB value, first scale all values such that they are in the interval [0:360). This value can then be used as the hue in the HSB model.
  
  Hint: The module Color8 defines a function `Hsb2Rgb` that converts a HSB color description into its corresponding RGB representation.

### Exercise 27

In this exercise, we apply the filters from the previous case study to the mandelbrot pictures. As before, we have provided stubs for the missing functionality. For this exercise these can be found in the file Stencil tier3.sac.

1. Implement the functions `apply`, `sobelEdges`, `gaussBlur`, and `gaussBlur25` as described in the previous case study.

2. The three filters described above only operate on a single channel, e.g., a gray-scale image. To lift these to colour images, implement a corresponding function for each filter that applies the filter to each colour separately.